
Tasque is an asynchronous, bounded-concurrency task queue for Elixir.
It lets you enqueue anonymous functions or MFA tuples, run them under a
supervised Task.Supervisor, and receive results back via standard OTP
messages.
It is a good fit when you need bounded parallelism, per-task timeouts,
and OTP-friendly result delivery without introducing a separate job
system.

Good use cases for this library include:
	Database queries
	Communications with external APIs or services

Why Tasque?
	Bounded concurrency with a simple FIFO queue
	Anonymous function and MFA task support
	Result delivery through regular OTP messages
	Per-task timeouts that cover both queue wait time and execution time
	Support for atom, {:global, term}, and {:via, module, term} queue names

Quick Start
Add a queue to your supervision tree:
children = [
  {Tasque, name: MyApp.Queue, max_concurrency: 10}
]

Supervisor.start_link(children, strategy: :one_for_one)
Then enqueue work and await the result:
{:ok, ref} = Tasque.queue_task(MyApp.Queue, fn -> expensive_work() end)
{:ok, result} = Tasque.await(ref)
You can also consume results directly from the caller mailbox:
{:ok, ref} = Tasque.queue_task(MyApp.Queue, fn -> String.upcase("hello") end)

receive do
  {:tasque_result, ^ref, {:ok, result}} -> result
  {:tasque_result, ^ref, {:exit, reason}} -> {:error, reason}
end
Task Formats
Tasks can be provided as either:
	A zero-arity function: fn -> :work end
	An MFA tuple: {String, :upcase, ["hello"]}

Invalid task formats return {:error, :invalid_task}.
Result Format
Every task result is delivered to the calling process as:
{:tasque_result, ref, outcome}
Where outcome is one of:
	{:ok, result} when the task completes successfully
	{:exit, reason} when the task crashes, exits, or times out

Timeouts
Per-task timeouts start when a task is enqueued, so they include both queue
wait time and execution time. If a timeout fires before dispatch, the task is
dropped from the queue and the caller receives {:exit, :timeout}. If it
fires while the task is running, the task process is terminated and the caller
receives the same timeout result.
This is separate from Tasque.await/2, whose timeout only controls how long
the caller waits for a result. An await/2 timeout does not cancel the task.
Naming
The queue :name supports the standard OTP naming forms:
	An atom such as MyApp.Queue
	A global name such as {:global, :my_queue}
	A via tuple such as {:via, Registry, {MyApp.Registry, "queue"}}

Tasque derives matching companion names for its supervisor processes using the
same naming strategy.
Examples
Different workloads often benefit from different concurrency limits. For
example, CPU-bound work is usually best capped near the number of schedulers,
while I/O-heavy work can often tolerate a higher limit:
children = [
  {Tasque, name: MyApp.CpuQueue, max_concurrency: System.schedulers_online()},
  {Tasque, name: MyApp.IoQueue, max_concurrency: 50}
]
You can then route work to the appropriate queue:
{:ok, image_ref} =
  Tasque.queue_task(MyApp.CpuQueue, fn -> render_thumbnail(image) end)

{:ok, api_ref} =
  Tasque.queue_task(MyApp.IoQueue, fn -> fetch_remote_profile(user_id) end, timeout: 5_000)

{:ok, thumbnail} = Tasque.await(image_ref)
{:ok, profile} = Tasque.await(api_ref)
MFA tasks work the same way:
{:ok, ref} = Tasque.queue_task(MyApp.IoQueue, {String, :upcase, ["hello"]})
{:ok, "HELLO"} = Tasque.await(ref)
Caveats
	Task results are delivered to the process that called Tasque.queue_task/3.
If that caller exits before the task finishes, the result message is sent to
a dead PID and is effectively lost.
	Tasque.await/2 only controls how long the caller waits for a result. If it
returns {:error, :timeout}, the task may still be queued or running, and
its eventual {:tasque_result, ref, outcome} message will still arrive in
the caller's mailbox.
	Per-task :timeout is different from await/2 timeout. The queue-enforced
timeout starts at enqueue time, includes queue wait time, and may expire
before the task is ever dispatched.

Installation
The package can be installed by adding tasque to your list of dependencies in mix.exs:
def deps do
  [
    {:tasque, "~> 1.0.0"}
  ]
end
Documentation can be found at https://hexdocs.pm/tasque.

An asynchronous, bounded-concurrency task queue for Elixir.
Tasque lets you enqueue work — anonymous functions or MFA tuples — and
execute it asynchronously under a supervised Task.Supervisor, with a
configurable ceiling on simultaneous tasks. Results are delivered back to
the caller via plain OTP messages, so you can choose between blocking
(await/2) and non-blocking (selective receive) consumption styles.
Quick Start
Add a Tasque instance to your application's supervision tree:
children = [
  {Tasque, name: MyApp.Queue, max_concurrency: 10}
]

Supervisor.start_link(children, strategy: :one_for_one)
Then enqueue work from anywhere:
{:ok, ref} = Tasque.queue_task(MyApp.Queue, fn -> expensive_work() end)

# Block until the result arrives (or timeout):
{:ok, result} = Tasque.await(ref)
Result Format
Every task result is delivered as a {:tasque_result, ref, outcome} message
to the process that called queue_task/3. The outcome is one of:
	Outcome	Meaning
	{:ok, result}	Task completed successfully and returned result
	{:exit, reason}	Task crashed, was killed, or timed out

When using await/2, the outcome tuple is returned directly.
Task Formats
Tasks can be provided as either a zero-arity anonymous function or an
MFA tuple. MFA tuples are normalized internally to fn -> apply(m, f, a) end.
# Anonymous function
Tasque.queue_task(queue, fn -> String.upcase("hello") end)

# MFA tuple
Tasque.queue_task(queue, {String, :upcase, ["hello"]})
Supervision Architecture
Each Tasque instance starts a small supervision subtree:
Tasque.Supervisor (:one_for_all)
├── Task.Supervisor   — runs tasks via async_nolink
└── Tasque.Queue      — GenServer managing the FIFO queue
The :one_for_all strategy ensures that if either child crashes, both
are restarted together, keeping the system in a consistent state. The
Task.Supervisor starts first so its name is available to Tasque.Queue
at init time.
Concurrency Control
When the number of in-flight tasks is below :max_concurrency, enqueued
work is dispatched immediately. Once the limit is reached, tasks wait in
a FIFO queue and are dispatched as running tasks complete, crash, or
time out — each of these events frees a concurrency slot.
Fault Isolation
Tasks run under Task.Supervisor.async_nolink/2, so a crashing task:
	Does not crash the queue GenServer
	Does not affect other concurrently running tasks
	Produces an {:exit, reason} result delivered to the original caller
	Frees its concurrency slot, allowing the next queued task to dispatch

Per-Task Timeouts
You can set a timeout on individual tasks at enqueue time:
Tasque.queue_task(queue, fn -> slow_work() end, timeout: 5_000)
The timeout starts when the task is enqueued, so it includes time spent
waiting in the queue as well as time spent running.
If the timeout fires while the task is still queued, the queue tombstones
it, delivers {:exit, :timeout} to the caller, and skips it during
dispatch.
If the timeout fires while the task is already running:
	The task process is killed (Task.Supervisor.terminate_child/2)
	{:exit, :timeout} is delivered to the caller
	The concurrency slot is freed

Pass timeout: :infinity to disable the per-task timeout. Otherwise,
:timeout must be a positive integer in milliseconds. If no :timeout
option is given, the task runs without a timeout.
Await timeout vs. task timeout
Tasque.await/2's timeout controls how long the caller is willing
to wait. It does not cancel the task — the task will continue
running, its result will still be delivered to the caller's mailbox,
and it will still occupy a concurrency slot.
The :timeout option on queue_task/3 is enforced by the queue
itself. If it fires while the task is already running, the task
process is terminated. If it fires while the task is still queued,
the task is dropped before dispatch.
Multiple Queues
You can run independent queues with different concurrency limits for
different workload categories:
children = [
  {Tasque, name: MyApp.CpuQueue, max_concurrency: System.schedulers_online()},
  {Tasque, name: MyApp.IoQueue,  max_concurrency: 50}
]
Each queue maintains its own task supervisor, dispatch loop, and
concurrency budget.
Caller Refs
The reference returned by queue_task/3 is a caller ref — a
make_ref() created by the queue GenServer and decoupled from the
internal Task ref used by the task supervisor. This means:
	The caller never holds a direct reference to the underlying task process
	Refs are unique per enqueue call, even if the same function is submitted twice
	The queue maintains the mapping from internal task refs to caller refs

Edge Cases
	Caller dies before task completes — the result message is sent
to a dead PID and silently dropped. The task still runs to completion
(or timeout) and the concurrency slot is freed normally.

	Await times out — await/2 returns {:error, :timeout} but
the task keeps running. The {:tasque_result, ref, outcome} message
will eventually arrive in the caller's mailbox. You can call await/2
again with the same ref to retrieve it, or use a selective receive.

          @spec await(ref :: reference(), timeout :: timeout()) ::
  {:ok, result :: any()} | {:exit, reason :: any()} | {:error, :timeout}


      


Block the calling process until the task identified by ref completes
or the timeout expires.
Returns the task outcome directly:
	{:ok, result} — task completed successfully
	{:exit, reason} — task crashed or was killed (including :timeout)
	{:error, :timeout} — the await timed out (task is still running)

The default timeout is 5000 ms.
Important
An await timeout does not cancel the task. The task continues
running, occupies its concurrency slot, and its result message will
still arrive in the caller's mailbox. If you need to enforce a hard
deadline, use the :timeout option on queue_task/3 instead.
Examples
iex> {:ok, _} = Tasque.start_link(name: Tasque.AwaitDoc, max_concurrency: 5)
iex> {:ok, ref} = Tasque.queue_task(Tasque.AwaitDoc, fn -> 42 end)
iex> Tasque.await(ref)
{:ok, 42}
With a custom timeout:
iex> {:ok, _} = Tasque.start_link(name: Tasque.AwaitTimeoutDoc, max_concurrency: 5)
iex> {:ok, ref} = Tasque.queue_task(Tasque.AwaitTimeoutDoc, fn -> Process.sleep(:infinity) end)
iex> Tasque.await(ref, 50)
{:error, :timeout}

Returns a child specification for starting a Tasque instance under a supervisor.
This is invoked automatically when you use {Tasque, opts} tuple syntax in
a supervision tree.
Options
	:name (required) — the name used to register the queue. Can be an
atom, a {:global, term} tuple, or a {:via, module, term} tuple.
Internal processes derive their names from this value.
For example, if name: MyApp.Queue is provided, the derived names
are MyApp.Queue.TaskSupervisor and MyApp.Queue.Supervisor.
For :global and :via names, Tasque derives matching companion
names using the same naming strategy.

	:max_concurrency — the maximum number of tasks that may execute
simultaneously. Defaults to 10.


Examples
As part of a supervision tree:
children = [
  {Tasque, name: MyApp.Queue, max_concurrency: 5}
]
Or build the spec manually:
Tasque.child_spec(name: MyApp.Queue, max_concurrency: 5)
#=> %{id: MyApp.Queue, start: {Tasque.Supervisor, :start_link, [[name: MyApp.Queue, max_concurrency: 5]]}, type: :supervisor}

          @spec queue_task(GenServer.server(), task(), keyword()) ::
  {:ok, ref :: reference()} | {:error, :invalid_task}


      


Enqueue a task for asynchronous execution. Returns immediately with
{:ok, ref}, where ref is a unique reference for tracking the result.
The result will be delivered as a {:tasque_result, ref, outcome} message
to the calling process. Use await/2 or a selective receive to consume it.
Task Formats
	Zero-arity function — fn -> :work end
	MFA tuple — {Module, :function, [args]}

Options
	:timeout — per-task timeout as a positive integer in milliseconds,
or :infinity. The timeout starts when the task is enqueued, so it
includes time spent waiting in the queue as well as time spent running.
If it fires while queued, the task is dropped before dispatch; if it
fires while running, the task process is killed. In either case,
{:exit, :timeout} is delivered. If omitted, no timeout is applied.

Returns
	{:ok, ref} on success.
	{:error, :invalid_task} if the task is neither a zero-arity function nor an MFA tuple.

Examples
iex> {:ok, _} = Tasque.start_link(name: Tasque.QueueTaskDoc, max_concurrency: 5)
iex> {:ok, ref} = Tasque.queue_task(Tasque.QueueTaskDoc, fn -> 1 + 1 end)
iex> is_reference(ref)
true
iex> Tasque.await(ref)
{:ok, 2}
With an MFA tuple:
iex> {:ok, _} = Tasque.start_link(name: Tasque.QueueTaskMfaDoc, max_concurrency: 5)
iex> {:ok, ref} = Tasque.queue_task(Tasque.QueueTaskMfaDoc, {String, :upcase, ["hello"]})
iex> Tasque.await(ref)
{:ok, "HELLO"}
With a per-task timeout:
iex> {:ok, _} = Tasque.start_link(name: Tasque.QueueTaskTimeoutDoc, max_concurrency: 5)
iex> {:ok, ref} = Tasque.queue_task(Tasque.QueueTaskTimeoutDoc, fn -> Process.sleep(:infinity) end, timeout: 50)
iex> receive do
...>   {:tasque_result, ^ref, outcome} -> outcome
...> after
...>   1_000 -> :never_arrived
...> end
{:exit, :timeout}

Starts a Tasque instance directly.
This is primarily useful for testing or starting a queue dynamically.
In production applications, it is heavily recommended to start Tasque
as part of your application's supervision tree instead of calling
start_link/1 directly.
See child_spec/1 for supported options.

Internal GenServer that manages the FIFO task queue, concurrency gating,
and dispatch loop for a single Tasque instance.
Internal module
This module is not part of the public API. Use the functions in Tasque
to interact with the queue. The implementation details documented here
are provided for contributors and the curious.
State
The GenServer maintains the following fields:
	:queue — an Erlang :queue of task entries waiting to be dispatched
	:pending_refs — a map of internal task_ref => entry for running tasks
	:queued_refs — a map of caller_ref => pid for waiting tasks
	:cancelled_refs — a map of caller_ref => true for tombstoned tasks
	:caller_to_task_ref — a map of caller_ref => task_ref for fast timeout lookups
	:max_concurrency — the upper bound on map_size(pending_refs)
	:task_supervisor — the registered name of the companion Task.Supervisor

Dispatch Algorithm
The private dispatch/1 function is called after every state-changing
event (enqueue, completion, crash, timeout). It greedily fills available
concurrency slots:
	If map_size(pending_refs) >= max_concurrency, return immediately
	If the queue is empty, return immediately
	Otherwise, dequeue the next entry, start it via
Task.Supervisor.async_nolink/2, record it in :pending_refs, and
recurse to fill any remaining slots

Message Protocol
Tasks are started with async_nolink, so results arrive as handle_info
messages:
	Message	Meaning	Action
	{task_ref, result}	Successful completion	Deliver {:ok, result}, demonitor, free slot
	{:DOWN, task_ref, :process, _, reason}	Task crashed	Deliver {:exit, reason}, free slot
	{:tasque_timeout, caller_ref}	Per-task timeout fired	If queued, tombstone and skip during dispatch; if running, terminate task; in both cases deliver {:exit, :timeout}

A catch-all handle_info/2 clause silently discards unexpected messages
(e.g., a late :DOWN arriving after a timeout has already handled the task).
Timeout Implementation
When a task with a :timeout option is enqueued, the queue schedules
a {:tasque_timeout, caller_ref} message to itself via
Process.send_after/3. If the task is still in the queue when the
timeout fires, it is tombstoned and skipped during dispatch. If it is
currently running, the queue attempts to terminate it and deliver a
timeout result. If the task has already completed by the time the
timeout message is handled, the normal completion or crash message wins.
If the task completes before the timer fires, the timer is cancelled with
Process.cancel_timer/1. Late timeout messages for already-completed
tasks are harmless no-ops.

Returns a specification to start this module under a supervisor.
See Supervisor.

Internal supervisor that manages the process pair for a single Tasque
instance: a Task.Supervisor and a Tasque.Queue GenServer.
Internal module
This module is not part of the public API. Use {Tasque, opts} in your
supervision tree instead of starting this module directly.
Supervision Strategy
Uses :one_for_all — if either the Task.Supervisor or the
Tasque.Queue crashes, both are restarted together. This is necessary
because:
	If the Task.Supervisor crashes, all in-flight tasks are lost and
the queue's :pending_refs map would reference dead processes
	If the Tasque.Queue crashes, the mapping between task refs and
callers is lost, so in-flight task results could never be delivered

Restarting both ensures the system returns to a clean, consistent state.
Child Order
The Task.Supervisor is started before the Tasque.Queue so that
its registered name is available during queue initialization.
Process Naming
Given a Tasque instance named MyApp.Queue, the following processes
are registered:
	Process	Registered Name
	Tasque.Supervisor	MyApp.Queue.Supervisor
	Task.Supervisor	MyApp.Queue.TaskSupervisor
	Tasque.Queue	MyApp.Queue

For {:global, term} and {:via, module, term} queue names, Tasque uses
the same naming strategy for its companion supervisor processes.

      start_link(opts)



        
          
        

    

  


  


  


        

      


  OEBPS/dist/epub-4WIP524F.js
(()=>{var s=document.querySelector.bind(document),o=document.querySelectorAll.bind(document);function r(e){document.readyState!=="loading"?e():document.addEventListener("DOMContentLoaded",e)}var l="hll";window.addEventListener("exdoc:loaded",t);function t(){o("[data-group-id]").forEach(e=>{e.addEventListener("mouseenter",i),e.addEventListener("mouseleave",i)})}function i(e){let n=e.currentTarget,a=e.type==="mouseenter",c=n.getAttribute("data-group-id");n.parentElement.querySelectorAll(`[data-group-id="${c}"]`).forEach(u=>{u.classList.toggle(l,a)})}r(()=>{t()});})();




OEBPS/assets/logo.png





